
SUMMARY 

Step change-point and slope change-point models in the independent Poisson sequence 

are developed based on accumulated and doubly-accumulated statistics. The method for 

the step change-point model developed in Section 2 is an alternative to the likelihood ratio 

test of Worsley (1986) and the algorithm for 𝑝-value calculation based on the first-order 

Markov property is the same as that given there. Different algorithms for the non-null 

distribution and inference on the change-point itself are, however, newly developed and a 

Pascal program is given in the Appendix. These methods are extended to the slope change-

point model in Section 3. The approach is essentially the same as that of Section 2 but the 

algorithm is now based on the second-order Markov property and becomes a little more 

complicated.  The Pascal program related to the slope change-point model is supported 

on the website, URL: https://corec.meisei-u.ac.jp/labs/hirotsu/. 

Key words: Convexity hypothesis; Markov property; Monotone hypothesis; Slope 

change-point model; Step change-point model 

1. Introduction 

Spontaneous reporting of adverse events due to a drug is collected daily, weekly 

or monthly by a medical organization such as PMDA (Pharmaceutical and 

Medical Device Agency of Japan), see Table 1, for example. These data are the 

number of events reported per a month from November 2003 to May 2010 at 

PMDA ; an independent Poisson sequence is assumed. It is important at PMDA 

to detect a significant change of time series in as short a time as possible.  

Table 1. Spontaneous reporting of adverse events per month at PMDA 

𝑘 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 

𝑦𝑘 1 4 1 1 1 1 3 0 4 1 3 0 2 4 3 

𝑘 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 

𝑦𝑘 3 2 4 1 4 1 4 2 1 2 2 1 0 1 5 

𝑘 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 

𝑦𝑘 1 4 1 4 2 3 7 3 3 4 1 5 4 5 6 

𝑘 46 47 48 49 50 51 52 53 54 55 56 57 58 59 60 

𝑦𝑘 2 4 9 3 4 1 1 6 3 5 8 1 1 6 3 

𝑘 61 62 63 64 65 66 67 68 69 70 71 72 73 74 75 

𝑦𝑘 3 1 2 3 1 3 4 3 3 5 2 2 0 4 4 

𝑘 76 77 78 79                       

𝑦𝑘 4 2 2 4                       

 

According to Worsley (1986) we consider a fixed segment of a Poisson 

sequence but the same statistic can be applied also sequentially with a slight 

modification. For such data, detecting an increasing tendency on the whole is of 

interest as well as detecting a step change-point. Then the maximal standardized 

accumulated statistic, max acc. 𝑡1, has been shown to be an appropriate statistic 

to detect both an increasing tendency and the change-point simultaneously, see 

Hirotsu and Marumo (2002) and Hirotsu (2013). It is based on the accumulated 

efficient score and belongs to the complete class of tests for the monotone 

alternative in the means of a Poisson sequence (Hirotsu, 1982). It is also 

asymptotically equivalent to the likelihood ratio test statistic to detect a step 

change-point in a Poisson sequence given by Worsley (1986). In the change-point 

analysis there are two parameters of interest, one is the change in mean and the 

other is the change-point itself. The algorithm for calculating 𝑝-values for testing 

a change is essentially the same as given in Worsley (1986) but the algorithm for 

the 𝑝-value for testing the null hypothesis of change-point itself is novel.  

An approach to the step change-point is nicely extended to the slope change-

point model and max acc. 𝑡2 has been proposed in Hirotsu et al. (2016). The 

probability calculations become a little harder now based on the second-order 

Markov property but the underlying idea is the same. 

Algorithms related to a step change-point are given in Chapter 2 and the 

related Pascal program is given in the Appendix. The Pascal program related to 

the slope change-point model is given at URL: https://corec.meisei-

u.ac.jp/labs/hirotsu/. 

2. Monotone Hypothesis and Step Change-point Model 

2.1. Model and Basic Idea 

We assume a sequence of variables 𝑦𝑖 which are distributed independently as 

Poisson random variables with distribution 𝑒−𝛬𝑖(𝛬𝑖)
𝑦𝑖/𝑦𝑖! of mean 𝛬i, 𝑖 =

1,… , 𝑎. Then for testing the null hypothesis 

𝐻𝑚0: 𝛬1 = ⋯ = 𝛬𝑎                                                                                                   (1) 

against a monotone alternative 

𝐻𝑚1: 𝛬1 ≤ ⋯ ≤ 𝛬𝑎, with at least one inequality strong,                              (2) 

max acc. 𝑡1 has been proposed in Hirotsu (1982) as one of the tests that belongs 

to the complete class of tests for the monotone hypothesis. It is defined by max 

acc. 𝑡1 = max(𝑡1,⋯ , 𝑡𝑎−1), where  

𝑡𝑘(𝑌𝑘) = 𝑡𝑘 = {(
𝑎−𝑘

𝑎𝑘
) �̂�}

−1/2

(�̂� −
𝑌𝑘

𝑘
) , 𝑘 = 1,… , 𝑎 − 1,                            (3) 

with 𝑌𝑘 = 𝑦1 + ⋯+ 𝑦𝑘 , �̂� = 𝑌𝑎/𝑎. 

The statistic 𝑌𝑘 is the accumulated efficient score since 𝒚 = (𝑦1, … , 𝑦𝑘)′ is an 

efficient score vector with respect to 𝛬 = (𝛬1,⋯ , 𝛬𝑎)′, where throughout this 

paper a prime implies a transpose of a vector or a matrix. Then 𝑡𝑘(𝑌𝑘) is the 

standardized version of 𝑌𝑘 under the null hypothesis 𝐻𝑚0. The accumulated 

statistic 𝑌𝑘 is shown to be also an efficient score with respect to parameter ∆ in 

the step change-point model, 

                  𝜃𝑖 = 𝑙𝑜𝑔 𝛬𝑖 = 𝜃, 𝑖 = 1,⋯ , 𝑘, 

𝑀𝑚𝑘:                                                                                                                                 (4) 

               𝜃𝑖 = 𝑙𝑜𝑔 𝛬𝑖 = 𝜃 + ∆, 𝑖 = 𝑘 + 1,… , 𝑎. 

Therefore the max acc. 𝑡1 is an appropriate test statistic also for the null 

hypothesis 

𝐻∆: ∆= 0, for all unknown 𝑘,                                                                                  (5) 

in change-point model (4). It should be noted that the model 𝑀𝑚𝑘 is a typical 

example of the monotone hypothesis satisfying 𝐻𝑚1. On the contrary, every 

monotone contrast 𝒄 = (𝑐1, ⋯ , 𝑐𝑎)′, 𝒄′𝒋 = 0, 𝑐1 ≤ ⋯ ≤ 𝑐𝑎 , 𝒋 = (1,⋯ ,1)′ can be 

expressed by a unique and positive linear combination of step change-point 

contrasts (𝑘 − 𝑎,⋯ , 𝑘 − 𝑎, 𝑘,⋯ , 𝑘) with the first 𝑘 elements 𝑘 − 𝑎 and the last 

𝑎 − 𝑘 elements 𝑘, 𝑘 = 1,… , 𝑎 − 1,thus suggesting a close relationship between 

the monotone hypothesis and the step change-point model (Hirotsu and Marumo, 

2002). The null hypotheses 𝐻𝑚0 (1) and 𝐻∆ (5) are of course equivalent.  

For the change-point model 𝑀𝑚𝑘(4) we are interested also in testing the null 

hypothesis that 𝐾 + 1 is the change-point, 

 𝐻0
𝐾+1: 𝑘 + 1 = 𝐾 + 1,                                                                                             (6) 

against the alternative hypothesis 

𝐻1
𝐾+1: 𝑘 + 1 ≠ 𝐾 + 1                                                                                               (7) 

asserting 𝐾 + 1 not to be a change-point. 

2.2. An Algorithm for Calculation of 𝒑-value under the Null Model  

𝑯𝒎𝟎 or 𝑯∆ 

First define the conditional probability given 𝑌𝑘 as 𝐹𝑘(𝑌𝑘  , 𝑡0) =  

Pr(𝑡1 < 𝑡0,⋯ , 𝑡𝑘 < 𝑡0|𝑌𝑘), 𝑘 = 1,… , 𝑎,where  𝑡𝑎(𝑌𝑎) is defined to be − ∞. 

Then we have a recursion formula:  

𝐹𝑘+1(𝑌𝑘+1, 𝑡0) =  𝑃𝑟(𝑡1 < 𝑡0, ⋯ , 𝑡𝑘 < 𝑡0, 𝑡𝑘+1 < 𝑡0|𝑌𝑘+1)    

        = ∑ 𝑃𝑟(𝑡1 < 𝑡0, ⋯ , 𝑡𝑘 < 𝑡0, 𝑡𝑘+1 < 𝑡0|𝑌𝑘 , 𝑌𝑘+1)𝑌𝑘
𝑓𝑘(𝑌𝑘|𝑌𝑘+1)            (8) 

          =  {
∑ 𝐹𝑘(𝑌𝑘 , 𝑡0)𝑌𝑘

𝑓𝑘(𝑌𝑘|𝑌𝑘+1)  𝑖𝑓 𝑡𝑘+1(𝑌𝑘+1) < 𝑡0 ,

0, 𝑜𝑡ℎ𝑒𝑟𝑤𝑖𝑠𝑒,
                                 (9)  

where 𝑓𝑘(𝑌𝑘|𝑌𝑘+1) is a conditional distribution of 𝑌𝑘 given 𝑌𝑘+1 and the range of 

𝑌𝑘 in the summation is obviously restricted to 0 ≤ 𝑌𝑘 ≤ 𝑌𝑘+1. In this case the 

conditional distribution is known to be a binomial distribution 

 𝑓𝑘(𝑌𝑘|𝑌𝑘+1) = (
𝑌𝑘+1

𝑌𝑘
) (

𝑘

𝑘+1
)

𝑌𝑘
(

1

𝑘+1
)

𝑌𝑘+1−𝑌𝑘
.  

Equation (8) is due to the law of total probability and equation (9) is due to 

Markov property of the sequence {𝑌𝑘}. Finally we have the 𝑝-value for the 

observed maximum 𝑡0 at the final step as 𝑝 = 1 − 𝐹𝑎(𝑌𝑎 , 𝑡0). It should be noted 

that the procedure converts the multiple summation into the repetition of a single 

summation, so that the calculation is feasible for large 𝑎. The Pascal program 

given in the Appendix can be applied to the 𝑝-value calculation. 

 

Example 1. Change-point analysis of Table 1. Max acc. 𝑡1 is applied with the 

estimate �̂� =2.835. The observed maximum is max 𝑡𝑘(𝑌𝑘) = 3.497 at April 

2006 (𝑘 = 29) with right one-sided 𝑝-value 0.0096 by the recursion formula (9). 

It should be noted that large 𝑡𝑘(𝑌𝑘) suggests a shift of mean between 𝑘 and 

 𝑘 + 1; we call this a change at 𝑘 + 1. Therefore, max acc. 𝑡1 suggests a shift of 

mean at 𝑘 + 1 = 30. Then it is interesting to confirm whether it changed to  

a decreasing tendency at some point after that time (𝑘 + 1 = 30). For this 

purpose, max acc. 𝑡2 of Section 3.2 can be applied. 

2.3. Power Calculation 

We fix the change-point model 𝑀𝑚𝑘 (4) at 𝑘 as one of the monotone hypotheses 

and calculate the power of max acc. 𝑡1 as a function of ∆.  Essentially the same 

recursion formula for the power is obtained as from the 𝑝-value calculation 

except that the conditional distribution is now  𝑓𝑘
∗(𝑌𝑘|𝑌𝑘+1). It is then an 

easy calculation to obtain the factorization of the joint conditional distribution 

given 𝑌𝑎 under model 𝑀𝑚𝑘 (4)  as 

 𝐺(𝒚 | 𝑌𝑎 , ∆) = ∏ 𝑓𝑖
∗(𝑌𝑖|𝑌𝑖+1)

𝑎−1
𝑖=1 ,  

where 

𝑓𝑖
∗(𝑌𝑖|𝑌𝑖+1) = (

𝑌𝑖+1

𝑌𝑖
) (

𝑖

𝑖+1
)

𝑌𝑖
(

1

𝑖+1
)

𝑌𝑖+1−𝑌𝑖
, 𝑖 = 1, … , 𝑘 − 1,  

𝑓𝑖
∗(𝑌𝑖|𝑌𝑖+1) = 𝐶𝑖+1

−1 (𝑌𝑖+1, ∆) (
𝑌𝑖+1

𝑌𝑖
) (

𝑖

𝑖+1
)

𝑌𝑖
(

1

𝑖+1
)

𝑌𝑖+1−𝑌𝑖
𝑒−𝑌𝑖∆, 𝑖 = 𝑘, 

𝑓𝑖
∗(𝑌𝑖|𝑌𝑖+1) = 𝐶𝑖+1

−1 (𝑌𝑖+1, ∆)𝐶𝑖(𝑌𝑖 , ∆) (
𝑌𝑖+1

𝑌𝑖
) (

𝑖

𝑖 + 1
)

𝑌𝑖

(
1

𝑖 + 1
)

𝑌𝑖+1−𝑌𝑖

, 

                                                                                                                  𝑖 = 𝑘 + 1,… , 𝑎 − 1 

with 𝐶𝑖(𝑌𝑖, ∆), 𝑘 + 1 ≤ 𝑖 ≤ 𝑎 as normalizing constants. The conditional 

distribution 𝑓𝑖
∗(𝑌𝑖|𝑌𝑖+1) reduces to the binomial distribution 𝑓𝑖(𝑌𝑖|𝑌𝑖+1) under 𝐻∆. 

The coefficients of the conditional distribution are determined recursively. For 

the recursion formula, however, the simpler form  

 𝑓𝑖
∗(𝑌𝑖|𝑌𝑖+1) = 𝐶𝑖+1

−1 (𝑌𝑖+1)𝐶𝑖(𝑌𝑖){(𝑌𝑖+1 − 𝑌𝑖)!}
−1, 𝑖 = 1, … , 𝑘 − 1, 

𝑓𝑘
∗(𝑌𝑘|𝑌𝑘+1) = 𝐶𝑘+1

−1 (𝑌𝑘+1, ∆)𝐶𝑘(𝑌𝑘)𝑒
−𝑌𝑘∆{(𝑌𝑘+1 − 𝑌𝑘)!}

−1, 𝑖 = 𝑘,                (10) 

𝑓𝑖
∗(𝑌𝑖|𝑌𝑖+1) = 𝐶𝑖+1

−1 (𝑌𝑖+1, ∆)𝐶𝑖(𝑌𝑖 , ∆){(𝑌𝑖+1 − 𝑌𝑖)!}
−1, 𝑖 = 𝑘 + 1,… , 𝑎 − 1, 

is made more convenient by slightly changing the definition of the coefficients 

𝐶𝑖. Then the coefficients 𝐶𝑖 are calculated recursively starting from 𝐶1 = 1/𝑌1! by 

𝐶𝑖+1(𝑌𝑖+1) = ∑ 𝐶𝑖(𝑌𝑖){(𝑌𝑖+1 − 𝑌𝑖)!}
−1

𝑌𝑖
, 𝑖 = 1, … , 𝑘 − 1,  

𝐶𝑘+1(𝑌𝑘+1, ∆) = ∑ 𝐶𝑘(𝑌𝑘)𝑒
−𝑌𝑘∆{(𝑌𝑘+1 − 𝑌𝑘)!}

−1
𝑌𝑘

, 𝑖 = 𝑘,                              (11) 

𝐶𝑖+1(𝑌𝑖+1, ∆) = ∑ 𝐶𝑖(𝑌𝑖 , ∆){(𝑌𝑖+1 − 𝑌𝑖)!}
−1

𝑌𝑖
, 𝑖 = 𝑘 + 1,… , 𝑎 − 1. 

Recursion formula (9) is valid as it is, by replacing the conditional distribution 

𝑓𝑖(𝑌𝑖|𝑌𝑖+1) by 𝑓𝑖
∗(𝑌𝑖|𝑌𝑖+1) and putting a critical point instead of t0, and gives a 

useful method for calculating the power at given ∆. Equation (10) shows also that 

the accumulated statistic 𝑌𝑘 is the efficient score with respect to ∆ for the step 

change-point model (4). It should be noted that this algorithm can be applied also 

to the 𝑝-value calculation by setting ∆= 0. This formula is different from that of 

Worsley (1986) who considers two independent Poisson sequences on both sides 

of the change-point conditionally given 𝑌𝑘 . Our formula extends more 

conveniently to the slope change-point model in Chapter 3. A Pascal program 

given in the Appendix can be applied both to the power and p−value calculation. 

2.4. Inference on the Change-point 

In Example 1 we are also interested in the confidence set for the change-point 

𝑘 + 1. As usual, this is obtained as the set of 𝐾 + 1 that are not rejected at level 

𝛼 by the test of the null hypothesis on the change-point 𝐻0
𝐾+1 (6) against the 

alternative hypothesis 𝐻1
𝐾+1 (7) asserting 𝐾 + 1 not to be a change-point. An 

appropriate test statistic is again max𝑘 𝑡𝑘(𝑌𝑘) of (3) but the maximization is with 

respect to 𝑘 = 1,… , 𝑎 − 1, 𝑘 ≠ 𝐾 and its null distribution should be defined 

under the null hypothesis 𝐻0
𝐾+1. This statistic is asymptotically equivalent to the 

likelihood ratio statistic used by Worsley (1986). In this case the null distribution 

contains a nuisance parameter ∆. However, according to Worsley, we can make 

the inference free from ∆ by conditioning on the sufficient statistic 𝑌𝐾 under 

𝐻0
𝐾+1. The conditional null distribution is most easily obtained by running the 

recursion formulae (9) and (11), fixing 𝑌𝐾 at the observed value. This is easily 

done by altering the inequality for restricting 𝑌𝐾 to the one point of the observed 

value at step 𝑖 = 𝐾 in running recursion formula (11).  Then, the confidence set 

eventually collects those 𝐾 + 1 for which 𝑡𝐾(𝑌𝐾) is sufficiently close to the 

observed max𝑘 𝑡𝑘(𝑌𝑘), 𝑘 = 1,… , 𝑎 − 1. Worsley (1986) proposed a different 

recursion formula considering independent Markov processes for both sides of 

the assumed change-point. However, our method is more convenient since the 

calculations of 𝑝-value, power and the confidence set of change-points are 

performed in the one program given in the Appendix. Our method is also 

systematically extended to the second order Markov sequence in Section 3. 

 

Example 2. Example 1 continued. We test the null hypotheses 𝐻0
𝐾+1 (6) for 𝐾 =

1,… , 𝑎 − 1, applying the recursion formulae (9) and (11), and collect those 

𝐾 + 1 with two-sided 𝑝-value larger than or equal to 0.10. Then the confidence 

set at confidence coefficient 0.90 is obtained as an interval 27 ≤ 𝐾 + 1 ≤ 43. 

3. Convexity Hypothesis and Slope Change-point Model 

3.1. Model and Basic Idea 

The idea of Section 2 is extended to the convexity hypothesis and slope change-

point model almost as it is. It should be noted here that the convexity hypothesis 

depends on the spacing of events, whereas monotonicity is a property 

independent of spacing. Therefore, we consider here a general case of unequal 

spacing of events and denote the time or location of the 𝑖th event by 𝑥𝑖. Then, the 

convexity hypothesis was introduced in Hirotsu and Marumo (2002) as 

           𝐻𝑐1: 𝑳𝑎
∗′𝜽 ≥ 0,   with at least one inequality strong,                          (12) 

where 𝑳𝑎
∗′ is a second-order differential matrix, defined by 

𝑳𝑎
∗′ = 

[
 
 
 
 
 

1

𝑥2−𝑥1

1

𝑥1−𝑥2
+

1

𝑥2−𝑥3

1

𝑥3−𝑥2
0 0 … 0

0
1

𝑥3−𝑥2

1

𝑥2−𝑥3
+

1

𝑥3−𝑥4

1

𝑥4−𝑥3
0 … 0

…

0 0 0 …
1

𝑥𝑎−1−𝑥𝑎−2

1

𝑥𝑎−2−𝑥𝑎−1
+

1

𝑥𝑎−1−𝑥𝑎

1

𝑥𝑎−𝑥𝑎−1]
 
 
 
 
 

(𝑎−2)×𝑎

.   

The null hypothesis Hc0 is defined by (12) with all the equalities, 𝐻c0: 𝑳𝑎
∗′𝜽 = 0. 

Corresponding to max acc. t1 (3) for the monotone hypothesis, the statistic max 

acc. t2 has been developed based on  (𝑳𝑎
∗′𝑳𝑎

∗ )−1𝑳𝑎
∗′𝒚 by Hirotsu et al. (2016). It 

is defined by max acc. 𝑡2 = max(𝑠1
∗, ⋯ , 𝑠𝑎−2

∗ ), where  

 𝑠𝑘
∗ = {𝑆𝑘 − 𝐸(𝑆𝑘)}/𝑉

1/2(𝑆𝑘) 

is the standardized version of 𝑆𝑘 under 𝐻𝑐0 with 

𝑆𝑘 = ∑ (𝑘
𝑖=1 𝑥𝑖+1 − 𝑥𝑖)𝑌𝑖 = (𝑥𝑘+1 − 𝑥1)𝑦1 + ⋯+ (𝑥𝑘+1 − 𝑥𝑘)𝑦𝑘 ,  𝑘 = 1,… , 𝑎 − 2.      (13) 

The statistic 𝑆𝑘 is a weighted doubly-accumulated statistic of 𝒚 = (𝑦1, ⋯ , 𝑦𝑎)′, 

which reduces to 

 𝑆𝑘 = ∑ 𝑌𝑖

𝑘

𝑖=1
= 𝑘𝑦1 + (𝑘 − 1)𝑦2 + ⋯+ 𝑦𝑘 ,  𝑘 = 1,… , 𝑎 − 2, 

in the case of equal spacing. The formula for calculating moments for 

standardization is given in Hirotsu et al. (2016).  

It has been shown by Hirotsu and Marumo (2002) that 

𝑳𝑎
∗  (𝑳𝑎

∗′𝑳𝑎
∗ )−1 = [𝒃1, 𝒃2, … , 𝒃𝑎−2], 

                      𝒃𝑘 = (𝑰 − 𝜫𝑩) (0, … ,0, 𝑥𝑘+2 − 𝑥𝑘+1, … , 𝑥𝑎 − 𝑥𝑘+1)
′, 

where 𝜫𝑩 = 𝑩(𝑩′𝑩)−𝟏𝑩′ is a projection matrix on to the column space of  

 𝑩 = (
1,⋯ , 1

𝑥1, ⋯ , 𝑥𝑎
)

′

. 

Then, 

𝑀𝑐𝑘:   𝜽 = (𝑩 𝒃𝑘)(𝛽0  𝛽1 𝑘)
′ = 𝑩(𝛽0 𝛽1)

′ + 𝒃𝑘∆𝑘 

is a slope change-point model with change ∆𝑘 at time point 𝑥𝑘+1 for   

𝑘 = 1,… , 𝑎 − 2, where a positive ∆𝑘 corresponds to the convexity hypothesis. 

More explicitly this can be rewritten as  

                   𝜃𝑖 = 𝑙𝑜𝑔 𝛬𝑖 = 𝛽0
∗ + ∆𝑘𝑥𝑘+1 + (𝛽𝑘

∗ − ∆𝑘)𝑥𝑖 , 𝑖 = 1, … , 𝑘 + 1 

 𝑀𝑐𝑘:                                                                                                                               (14) 

               𝜃𝑖 = 𝑙𝑜𝑔 𝛬𝑖 = 𝛽0
∗ + 𝛽𝑘

∗𝑥𝑖 , 𝑖 = 𝑘 + 2,… , 𝑎. 

in a form similar to 𝑀𝑚𝑘 (4). The null model 𝐻𝑐0: 𝑳𝑎
∗′𝜽 = 0 is obviously 

equivalent to a linear regression model 𝐻𝑐0:  𝜃𝑖 = 𝛽0 + 𝛽1𝑥𝑖. The sufficient 

statistics under 𝐻𝑐0 are obviously  

   𝑌𝑎 = 𝑦1 + 𝑦2 + ⋯+ 𝑦𝑎  𝑎𝑛𝑑  𝑇𝑎 = 𝑥1𝑦1 + 𝑥2𝑦2 + ⋯ + 𝑥𝑎𝑦𝑎 , 

and a similar test is constructed conditionally given (𝑌𝑎 ,  𝑇𝑎). It is easily shown 

by derivation of the log likelihood function with respect to ∆𝑘under the model 

𝑀𝑐𝑘 (14) that the statistic 𝑆𝑘 is an efficient score for testing the slope change-

point model (14). Thus max acc. t2 is an appropriate test statistic simultaneously 

for the convexity and the slope change-point hypotheses. 

For the change-point model 𝑀𝑐𝑘 we are also interested in testing the null 

hypothesis that 𝑥𝐾+1 is the change-point, 

 𝐻0
𝑥𝐾+1: 𝑥𝑘+1 = 𝑥𝐾+1,                                                                                             (15) 

against the alternative hypothesis, 

𝐻1
𝑥𝐾+1: 𝑥𝑘+1 ≠ 𝑥𝐾+1                                                                                              (16) 

asserting 𝑥𝐾+1 not to be a change-point.   

3.2. An Algorithm for 𝒑-value Calculation under the Null Model 𝑯𝐜𝟎  

A recursion formula for the 𝑝-value has been obtained by Hirotsu et al. (2016) 

based on the second-order Markov property of the sequence {𝑆𝑘}. Let us define 

the conditional probability 

𝐹𝑘(𝑆𝑘−1, 𝑆𝑘 , 𝑑) = 𝑃𝑟(𝑠1
∗ < 𝑑,… , 𝑠𝑘

∗ < 𝑑|𝑆𝑘−1, 𝑆𝑘), 

                             = 𝑃𝑟(𝑆1 < 𝑑1
∗, … , 𝑆𝑘 < 𝑑𝑘

∗ |𝑆𝑘−1, 𝑆𝑘), 𝑘 = 2,… , 𝑎,              (17) 

where 𝑑𝑘
∗ = 𝐸(𝑆𝑘) + 𝑉1/2(𝑆𝑘)𝑑,  𝑘 = 1,… , 𝑎 − 2 and 𝑠𝑎−1

∗ , 𝑠𝑎
∗ are defined as 

−∞, although their conditional variances are zero so that the inequality always 

holds. It should be noted that the recursion formula (17) is given in terms of 𝑆𝑘, 

since the distribution theory has been obtained in terms of 𝑆𝑘 by Hirotsu et al. 

(2016). Correspondingly, 𝑑𝑎−1
∗  and 𝑑𝑎

∗  are set to 𝑆𝑎−1 + 𝛿 and 𝑆𝑎 + 𝛿, 

respectively, with 𝛿 a positive small number, where 𝑆𝑎−1 and 𝑆𝑎 are defined as 

𝑆𝑎−1 = 𝑥𝑎𝑌𝑎 − 𝑇𝑎, 𝑆𝑎 = 𝑆𝑎−1 + (𝑥𝑎+1 − 𝑥𝑎)𝑌𝑎 with 𝑥𝑎+1 an arbitrary number 

larger than 𝑥𝑎. It should be noted that 𝑆𝑎−1 is an extension of the definition (13) 

to 𝑘 = 𝑎 − 1, but it cannot be done for 𝑆𝑎 without introducing a hypothetical 

value 𝑥𝑎+1. For notational convenience anyway, 𝑆𝑎−1 and 𝑆𝑎 are employed as 

conditioning variables instead of 𝑌𝑎 and 𝑇𝑎. Also, it is sometimes convenient to 

use 𝑆𝑎−1 and 𝑌𝑎 as conditioning variables. The one-to-one correspondence among 

the set of variables (𝑌𝑎 , 𝑇𝑎), (𝑆𝑎−1, 𝑆𝑎), and (𝑆𝑎−1, 𝑌𝑎) is obvious. Then,  

a recursion formula for 𝐹𝑘 is obtained as 

𝐹𝑘+1(𝑆𝑘 , 𝑆𝑘+1, 𝑑) = 𝑃𝑟(𝑆1 < 𝑑1
∗, … , 𝑆𝑘 < 𝑑𝑘

∗ , 𝑆𝑘+1 < 𝑑𝑘+1
∗ |𝑆𝑘 , 𝑆𝑘+1) 

 = ∑ 𝑃𝑟(𝑆1 < 𝑑1
∗ , … , 𝑆𝑘 < 𝑑𝑘

∗ , 𝑆𝑘+1 < 𝑑𝑘+1
∗ |𝑆𝑘−1, 𝑆𝑘 , 𝑆𝑘+1) × 𝑓𝑘−1(𝑆𝑘−1|𝑆𝑘 , 𝑆𝑘+1)𝑆𝑘−1

        (18) 

       = {
𝑃𝑟(𝑆1 < 𝑑1

∗ , … , 𝑆𝑘 < 𝑑𝑘
∗ |𝑆𝑘−1, 𝑆𝑘) × 𝑓𝑘−1(𝑆𝑘−1|𝑆𝑘 , 𝑆𝑘+1)    𝑖𝑓 𝑆𝑘+1 < 𝑑𝑘+1 

∗ ,

0,       𝑜𝑡ℎ𝑒𝑟𝑤𝑖𝑠𝑒,
                          (19) 

where 𝑓𝑘(𝑆𝑘|𝑆𝑘+1, 𝑆𝑘+2) is the conditional distribution of 𝑆𝑘 given 𝑆𝑘+1 and 

𝑆𝑘+2. Equation (18) is due to the law of total probability, and equation (19) is due 

to the second-order Markov property of 𝑆𝑘. Thus, essentially the recursion 

formula is obtained as 

𝐹𝑘+1(𝑆𝑘 , 𝑆𝑘+1, 𝑑) = ∑ 𝐹𝑘(𝑆𝑘−1, 𝑆𝑘 , 𝑑) × 𝑓𝑘−1(𝑆𝑘−1|𝑆𝑘 , 𝑆𝑘+1)
𝑆𝑘−1

. 

There is no difficulty in extending the formula to 𝐹𝑎 and the 𝑝-value of the 

observed maximum 𝑠𝑚
∗  is obtained at the final step by 

𝑝 = 1 − 𝐹𝑎(𝑆𝑎−1, 𝑆𝑎 , 𝑑) 𝑎𝑡 𝑑 = 𝑠𝑚
∗ . 

It should be noted again that the procedure converts the multiple summation 

into the repetition of a single summation, so that the calculation is feasible for 

large 𝑎. 

The argument for the recursion formula above is similar to that of Section 2.2 

except that the conditioning variables become two, reflecting the second order 

Markov property. In this case, however, the conditional distribution 

𝑓𝑘(𝑆𝑘|𝑆𝑘+1, 𝑆𝑘+2) is unknown and needs to be calculated. 

3.3. Derivation and Calculation of the Conditional Distribution 

𝒇𝒌(𝑺𝒌|𝑺𝒌+𝟏, 𝑺𝒌+𝟐) 

The joint conditional probability of 𝒚 given {𝑌𝑎 , 𝑇𝑎} is factorized as 

 𝐺(𝒚|𝑌𝑎, 𝑇𝑎) = ∏ 𝑓𝑘(𝑆𝑘|𝑆𝑘+1, 𝑆𝑘+2)
𝑎−2
𝑘=1  

in terms of 𝑆𝑘 due to the second-order Markov property of 𝑆𝑘 . Then, the 𝑘th 

conditional distribution should be written in the form 

𝑓𝑘(𝑆𝑘|𝑆𝑘+1, 𝑆𝑘+2) = 𝐶𝑘+1
−1 (𝑆𝑘+1, 𝑆𝑘+2)𝐶𝑘(𝑆𝑘 , 𝑆𝑘+1) 

 

    × {(
𝑆𝑘+2 − 𝑆𝑘+1

𝑥𝑘+3 − 𝑥𝑘+2

−
𝑆𝑘+1 − 𝑆𝑘

𝑥𝑘+2 − 𝑥𝑘+1

) !}
−1

,  𝑘 = 1,… , 𝑎 − 2,       (20) 

where 𝐶𝑘+1 is the normalizing constant and (
𝑆𝑘+2−𝑆𝑘+1

𝑥𝑘+3−𝑥𝑘+2
−

𝑆𝑘+1−𝑆𝑘

𝑥𝑘+2−𝑥𝑘+1
) is 

nothing but 𝑦𝑘+2. 

The initial constant is defined as 

 𝐶1(𝑆1, 𝑆2) = {(
𝑆1

𝑥2 − 𝑥1

) !}
−1

× {(
𝑆2 − 𝑆1

𝑥3 − 𝑥2

−
𝑆1

𝑥2 − 𝑥1

) !}
−1

. 

It should further be noted that in equation (20) the random variable 𝑆𝑘 is also 

included as the conditioning variable in the normalizing constant 𝐶𝑘(𝑆𝑘 , 𝑆𝑘+1) of 

the previous step. Starting from 𝐶1, all the 𝐶𝑘 can be calculated recursively by the 

equation 

𝐶𝑘+1(𝑆𝑘+1, 𝑆𝑘+2) = ∑ 𝐶𝑘(𝑆𝑘 , 𝑆𝑘+1)
𝑆𝑘

× {(
𝑆𝑘+2 − 𝑆𝑘+1

𝑥𝑘+3 − 𝑥𝑘+2

−
𝑆𝑘+1 − 𝑆𝑘

𝑥𝑘+2 − 𝑥𝑘+1

) !}
−1

 .                          (21) 

Then, at the final step the overall normalizing constant 𝐶𝑎−1(𝑆𝑎−1, 𝑆𝑎) is obtained 

and the distribution 𝐺(𝒚) is determined in terms of 𝑆𝑘, 𝑘 = 1,… , 𝑎 − 2. Note that 

𝐶𝑎−1(𝑆𝑎−1, 𝑆𝑎) is well defined, since (𝑆𝑎 − 𝑆𝑎−1)/(𝑥𝑎+1 − 𝑥𝑎) is simply 𝑌𝑎, but 

in what follows the notation 𝐶𝑎−1(𝑆𝑎−1, 𝑌𝑎) is employed instead of 

𝐶𝑎−1(𝑆𝑎−1, 𝑆𝑎). For executing the recursion formula efficiently, several 

inequalities have been introduced in Hirotsu et al. (2016) and we give a summary 

in the following: 

Absolute:  

Relative: 

 
𝑥𝑘+1 − 𝑥1

𝑥𝑘 − 𝑥1
𝑆𝑘−1 ≤ 𝑆𝑘 ≤

1

𝑥𝑎 − 𝑥𝑘

{(𝑥𝑎 − 𝑥𝑘+1)𝑆𝑘−1 + (𝑥𝑘+1 − 𝑥𝑘)𝑆𝑎−1},  

                                                                                                                𝑘 = 2,… , 𝑎 − 2. 

The absolute inequality gives restrictions on 𝑆𝑘 in terms of 𝑆𝑎−1 and 𝑌𝑎, which 

are constants, and the relative one is useful for the bottom-up procedure to 

  max{0, 𝑆𝑎−1 − (𝑥𝑎 − 𝑥𝑘+1)𝑌𝑎 } ≤ 𝑆𝑘 ≤
𝑥𝑘+1 − 𝑥1

𝑥𝑎 − 𝑥1
𝑆𝑎−1, 𝑘 = 1,… , 𝑎 − 2.  

construct conditional probabilities giving the possible range of 𝑆𝑘 in terms of 

𝑆𝑘−1. This inequality can be rewritten as 

               
1

𝑥𝑎−𝑥𝑘+1
{(𝑥𝑎 − 𝑥𝑘)𝑆𝑘 − (𝑥𝑘+1 − 𝑥𝑘)𝑆𝑎−1} ≤ 𝑆𝑘−1 ≤

𝑥𝑘−𝑥1

𝑥𝑘+1−𝑥1
𝑆𝑘 ,  𝑘 = 2,… , 𝑎 − 2,  

which is useful for defining the range of 𝑆𝑘−1 based on 𝑆𝑘.  

 

Example 3: Example 1 continued. In this example we are interested in testing a 

downturn tendency and therefore we perform a concavity test based on −𝑠𝑘
∗. 

Applying the recursion formula (19) for concavity test, a downturn is detected at 

October 2007 (𝐾 + 1 = 48) with observed maximum −𝑠𝑚
∗ = 2.858 and one-

sided 𝑝-value 0.0093 by the aid of recursion formula (21). 

3.4. Power Calculation 

We fix the change-point model 𝑀𝑐𝑘 (14) at 𝑘 and calculate the power of max acc. 

𝑡2 as a function of ∆𝑘. An essentially identical recursion formula is obtained as 

for the 𝑝-value calculation except for the conditional distribution 

𝑓𝑘(𝑆𝑘|𝑆𝑘+1, 𝑆𝑘+2). Now, it is an easy calculation to obtain the joint conditional 

distribution given (𝑌𝑎 , 𝑇𝑎) under 𝑀𝑐𝑘 in terms of 𝑆𝑘 as 

                 𝐺(𝒚|𝑌𝑎, 𝑇𝑎 , ∆𝑘) = 𝐶−1(𝑌𝑎, 𝑇𝑎, ∆𝑘)∏ {(
𝑆𝑖+2−𝑆𝑖+1

𝑥𝑖+3−𝑥𝑖+2
−

𝑆𝑖+1−𝑆𝑖

𝑥𝑖+2−𝑥𝑖+1
) !}

−1

𝑒𝑆𝑘∆𝑘𝑎−2
𝑖=1 .                        (22) 

Equation (22) shows again that the accumulated statistic 𝑆𝑘 is the efficient 

score with respect to ∆𝑘 for the slope change-point model (14). The factorization 

of 𝐺(𝒚|𝑌𝑎 , 𝑇𝑎 , ∆𝑘) is obtained by the same idea as equation (10) of Section 2.3, 

that is,  𝐺(𝒚│𝑌𝑎 , 𝑇𝑎 , ∆𝑘) = ∏ 𝑓𝑖
∗(𝑆𝑖|𝑆𝑖+1, 𝑆𝑖+2)

𝑎−2
𝑖=1 , where 

           𝑓𝑖
∗(𝑆𝑖|𝑆𝑖+1, 𝑆𝑖+2) = 𝐶𝑖+1

−1 (𝑆𝑖+1, 𝑆𝑖+2)𝐶𝑖(𝑆𝑖 , 𝑆𝑖+1) {(
𝑆𝑖+2−𝑆𝑖+1

𝑥𝑖+3−𝑥𝑖+2
−

𝑆𝑖+1−𝑆𝑖

𝑥𝑖+2−𝑥𝑖+1
) !}

−1

, 

 1 ≤ 𝑖 ≤ 𝑘 − 1, 

             𝑓𝑘
∗(𝑆𝑘|𝑆𝑘+1, 𝑆𝑘+2) = 𝐶𝑘+1

−1 (𝑆𝑘+1, 𝑆𝑘+2, ∆𝑘)𝐶𝑘(𝑆𝑘, 𝑆𝑘+1) {(
𝑆𝑘+2−𝑆𝑘+1

𝑥𝑘+3−𝑥𝑘+2
−

𝑆𝑘+1−𝑆𝑘

𝑥𝑘+2−𝑥𝑖+1
) !}

−1
𝑒𝑆𝑘∆𝑘 ,   

𝑖 = 𝑘, 

               𝑓𝑖
∗(𝑆𝑖|𝑆𝑖+1, 𝑆𝑖+2) = 𝐶𝑖+1

−1 (𝑆𝑖+1, 𝑆𝑖+2, ∆𝑘)𝐶𝑖(𝑆𝑖 , 𝑆𝑖+1, ∆𝑘) {(
𝑆𝑖+2−𝑆𝑖+1

𝑥𝑖+3−𝑥𝑖+2
−

𝑆𝑖+1−𝑆𝑖

𝑥𝑖+2−𝑥𝑖+1
) !}

−1
, 

𝑘 + 1 ≤ 𝑖 ≤ 𝑎 − 2. 

We can calculate the normalizing constants 𝐶𝑖(𝑆𝑖, 𝑆𝑖+1) recursively just as in 

equation (21), noting the change of kernel for calculating 

𝐶𝑘+1(𝑆𝑘+1, 𝑆𝑘+2, ∆𝑘) at 𝑖 = 𝑘. The constant 𝐶(𝑌𝑎 , 𝑇𝑎 , ∆𝑘) is obtained at the final 

step as 𝐶𝑎−1(𝑆𝑎−1, 𝑆𝑎 ,𝑘. The initial value is the same as (21). If the change-

point is at 𝑥 = 𝑥2, then the recursion formula starts from the second equation.  

3.5. Inference on the Change-point 

The confidence set of the change-point 𝑥𝑘+1 is obtained as the set of 𝑥𝐾+1 that 

are not rejected at level 𝛼 by the test of the null hypothesis on the change-point 

𝐻0
𝑥𝐾+1  (15) against the alternative hypothesis 𝐻1

𝑥𝐾+1 (16) asserting 𝑥𝐾+1 not to be 

a change-point. An appropriate test statistic is again the maximal standardized 

statistic of 𝑆𝑘 in Section 3.1, but the maximization is with respect to  

𝑘 = 1,… , 𝑎 − 2, 𝑘 ≠ 𝐾 and its null distribution should be defined under the null 

hypothesis 𝐻0
𝑥𝐾+1. Again, to be free from the nuisance parameter ∆𝐾, we make a 

conditional inference given the sufficient statistic 𝑆𝐾 under 𝐻0
𝑥𝐾+1  .  

The conditional null distribution is most easily obtained by running the recursion 

formula of Section 3.4 fixing 𝑆𝐾 at the observed value. This is just as in Section 

2.4. 

 

Example 4 Example 1 continued. In this example we are interested in testing a 

downturn tendency and therefore we perform a concavity test based on −𝑠𝑘
∗. 

Applying the recursion formula for 𝐾 = 1,… . , 𝑎 − 2, for the concavity test, the 

confidence set at confidence coefficient 0.90 is obtained as an interval 

 35 ≤ 𝑥𝐾+1 ≤ 58. 

4. Concluding Remarks 

The calculations of 𝑝-value, power and the confidence set of step change-points 

are performed in one Pascal program, which is given in the Appendix. The 

calculations for the slope change-point model are almost parallel to those for the 

step change-point model but are more complicated. Therefore, the related 

program becomes too long to present here but is available at URL: 

https://corec.meisei-u.ac.jp/labs/hirotsu/. 

APPENDIX 

program Max_acc_t1_Poisson(input, output); 

 

{$APPTYPE CONSOLE} 

 

uses SysUtils, Math; 

 

{ ------------------------------------------------------------------- } 

{   max acc t1 for Poisson sequence (step change-point model)         } 

{     1.p-value  2.power  3.confidence set                            } 

{ ------------------------------------------------------------------- } 

{   This program can be compiled by Delphi(R).                        } 

{   The latest version and the executable program are available at    } 

{      https://corec.meisei-u.ac.jp/labs/hirotsu/                     } 

{ ------------------------------------------------------------------- } 

{   Critical points for power calculation are available at the site.  } 

{ ------------------------------------------------------------------- } 

{                  Notes and Limitations                              } 

{ ------------------------------------------------------------------- } 

{   Global variables C1 & C2 are used to avoid stack overflow.        } 

{   Maximum length of Poisson sequence is 100.                        } 

{   Extended precision is used                                        } 

{   & the maximum number of total frequency is 1754.                  } 

{      (If you use double precision, max number is 170.)              } 

{ ------------------------------------------------------------------- } 

{ --- Declarations -------------------------------------------------- } 

 

  const  max_sequence  =                 100; 

         max_factorial =                1754;      { max of factorial } 

         epsilon       =           0.0000001; 

         infinity      =  1.797693134862E308; 

 

  type array_type        = array[0..max_sequence]  of integer; 

       array_type_R      = array[0..max_sequence]  of extended; 

       array_type_long_R = array[0..max_factorial] of extended; 

 

       Ck_type = record 

                   p        : extended; 

                   feasible : boolean; 

                 end; 

 

       option_type = ( get_p_value, get_power, get_CI ); 

 

  var  fi,   fo  : text; 

       filename  : string; 

       option    : option_type; 

 

       y         : array_type; 

       C1,  C2   : array[0..max_factorial] of Ck_type; { global var } 

       LFact     : array_type_long_R; 

 

       J, 

       K0,  n_option, 

       a,   Ya, 

       change_point   : integer; 

 

       p,   delta, 

       critical_point : extended; 

 

  label last_line; 

 

{ ------------------------------------------------------------------- } 

     function Imin2( x1, x2 : integer ) : integer; 

{ ------------------------------------------------------------------- } 

     begin 

       if x1 < x2 then Imin2 := x1 

                  else Imin2 := x2; 

     end; 

 

{ ------------------------------------------------------------------- } 

     function feasible( Y1, Y2 : integer ) : boolean; 

{ ------------------------------------------------------------------- } 

     var yk : integer; 

 

     begin 

       yk := Y2 - Y1; 

 

       if (yk >= 0) then feasible := True 



 

 

 

 

An algorithm for a new method of change-point analysis                   17 

                    else feasible := False; 

     end; 

 

{ ------------------------------------------------------------------- } 

     function fk( k,  Y1,  Y2, 

                  change_point : integer; 

                  delta        : extended ) : extended; 

{ ------------------------------------------------------------------- } 

{         NB: C1 & C2 are gloabal variables                           } 

{ ------------------------------------------------------------------- } 

     var yk : integer; 

         p  : extended; 

 

     begin 

       yk := Y2 - Y1; 

 

       if feasible(Y1, Y2) and ((C1[Y1].feasible and C2[Y2].feasible)) then 

         p := exp( Ln(C1[Y1].p) - Ln(C2[Y2].p) - LFact[yk] ) 

       else 

         p := 0; 

 

       if (k=change_point) then fk := p * exp(-delta*Y1) 

                           else fk := p; 

     end; 

 

{ ------------------------------------------------------------------- } 

     function get_F( a, K0, 

                     change_point   : integer; 

                     critical_point, 

                     delta          : extended; 

                     y              : array_type ) : extended; 

{ ------------------------------------------------------------------- } 

  var k,   J, 

      Y1,  Y2,   Ya, 

      low_Y1, up_Y1, 

      low_Y2, up_Y2, 

      max_k0          : integer; 

 

      max_t0,     t, 

      lambda_hat, sum : extended; 

 

      Yk,  low,  up   : array_type; 

      E0,  V0,   t0   : array_type_R; 

      Fp1, Fp2        : array_type_long_R; 

 

{ ----  Start of get_F  --------------------------------------------- } 

 

  begin 

    for J:=0 to max_factorial do begin 

      C1[J].p:=0;  C1[J].feasible:=False;  Fp1[J]:=0; 

      C2[J].p:=0;  C2[J].feasible:=False;  Fp2[J]:=0; 

    end; 

{ --- get accumulated sum Yk's of the observed sequence y ----------- } 

 

                     Yk[0] := 0; 

    for k:=1 to a do Yk[k] := Yk[k-1] + y[k]; 

                     Ya    := Yk[a]; 

 

{ --- set the range of Yk's ----------------------------------------- } 

 

    for k:=1 to (a-1) do begin 

      low[k] :=  0; 

      up[k]  := Ya; 

    end; 

 

    low[a] := Ya; 

    up[a]  := Ya; 

 

    if (K0>0) then begin 

      low[K0] := Yk[K0]; 

      up[K0]  := Yk[K0]; 

    end; 

 

{ --- calculate t0 based on the observed sequence ------------------- } 

 

    if (change_point=0) and (K0<=1) then begin 

      writeln( fo ); 

      writeln( fo, 'k,y[k],Yk[k],E[t],V[t],t[k]' ); 

 

      writeln; 

      writeln( '     k     y[k]    Yk[k]     E[t]        V[t]        t[k]' ); 

      writeln( '------------------------------------------------------------' ); 

    end; 

 

    max_t0     := -infinity; 

    max_k0     := 0; 

    lambda_hat := Ya/a; 

 

    if (change_point>0) then max_t0 := critical_point 

    else begin 

      for k:=1 to (a-1) do begin 

        E0[k] := lambda_hat - Yk[k]/k; 

        V0[k] := (1/k-1/a)*(lambda_hat); 

        t0[k] := E0[k] / sqrt(V0[k]); 

 

        if (t0[k] > max_t0) then begin 

          max_t0 := t0[k]; 

          max_k0 := k; 

        end; 

 

        if (change_point=0) and (K0<=1) then begin 

          writeln( fo,  k:8, ',', 

                        y[k]:4,    ',', 

                        Yk[k]:6,   ',', 

                        E0[k]:20:8,',', 

                        V0[k]:20:8,',', 

                        t0[k]:20:8      ); 

          writeln(      k:6,       '  ', 

                        y[k]:6,    '  ', 

                        Yk[k]:6,   '  ', 

                        E0[k]:10:4,'  ', 

                        V0[k]:10:4,'  ', 

                        t0[k]:10:4      ); 

        end; 

      end; 

    end; 

 

    max_t0 := max_t0 - epsilon; 

 

    if (change_point=0) and (K0<=1) then begin 

      writeln( fo ); 

      writeln( fo,  'max_t0:,',   max_t0:10:6, ',max at K:,', max_k0:6    ); 

      writeln( fo ); 

 

      writeln( '------------------------------------------------------------' ); 

      writeln; 

      writeln( 'max_t0 =', max_t0:10:4, ' at K = ', max_k0:3 ); 

      writeln; 

    end; 

 

{ --- get C(1)  ----------------------------------------------------- } 

 

    for Y1 := low[1] to up[1] do begin 

      C1[Y1].p        := exp( -LFact[Y1] ); 

      C1[Y1].feasible := True; 

 

      if (K0=1) then t := -infinity 

                else t := (lambda_hat - Y1/1) / sqrt((1/1- 1/a)*lambda_hat); 

 

      if (t<max_t0) then Fp1[Y1] := 1 

                    else Fp1[Y1] := 0; 

    end; 

 

{ --- get C(k+1) and then F(Y[k+1],t0) (k=1,..,(a-1)) --------------- } 

 

    for k:=1 to (a-1) do begin              { --- start of loop k --- } 

 

  { ---  Step 1 : get ranges of Y[k] and Y[k+1]  --- } 

 

      low_Y1 := low[k];    up_Y1  := up[k]; 

      low_Y2 := low[k+1];  up_Y2  := up[k+1]; 

 

  { ---  Step 2 : calculate C(k+1)  --- } 

 

      for Y2 := low_Y2 to up_Y2 do begin 

        sum := 0; 

 

        for Y1 := low_Y1 to up_Y1 do begin 

          if feasible(Y1, Y2) and C1[Y1].feasible then begin 

            C2[Y2].feasible := True; 

 

            if (k=change_point) then 

              sum := sum + exp( Ln(C1[Y1].p) - LFact[Y2-Y1]) * exp(-delta*Y1) 

            else 

              sum := sum + exp( Ln(C1[Y1].p) - LFact[Y2-Y1] ); 

          end; 

        end; 

 

        C2[Y2].p := sum; 

      end; 

 

  { ---  Step 3 : calculate F(Y[k+1],t0)  --- } 

 

      for Y2 := low_Y2 to up_Y2 do begin 

        if ((k+1)=a) or ((k+1)=K0) then 

          t := -infinity 

        else  

          t := (lambda_hat - Y2/(k+1)) / sqrt((1/(k+1)- 1/a)*lambda_hat); 

 

        if (t<max_t0) then begin 

          sum := 0; 

 

          for Y1 := low_Y1 to up_Y1 do 

            if feasible(Y1, Y2) then 

              sum := sum + Fp1[Y1] * fk( k, Y1, Y2, change_point, delta ); 

 

          Fp2[Y2] := sum; 

        end else 

          Fp2[Y2] := 0; 

      end; 

 

  { --- Step 4 : copy C(k+1) & Fp(k+1) for the next iteration --- } 

 

      if ((k+1) < a) then 

        for J:=0 to max_factorial do begin 

          C1[J].p:=C2[J].p; C1[J].feasible:=C2[J].feasible; Fp1[J]:=Fp2[J]; 

          C2[J].p:=0;       C2[J].feasible:=False;          Fp2[J]:=0; 

        end; 

    end;                                      { --- end of loop k --- } 

 

    get_F := Fp2[Ya]; 

  end;                                         { --- end of get_F --- } 

 

{ ------  M A I N    P A R T  --------------------------------------- } 

 

begin 

  K0             := 0; 

  delta          := 0; 

  change_point   := 0; 

  critical_point := -infinity; 

 



 

 

 

 

An algorithm for a new method of change-point analysis                   21 

  for J:=0 to max_sequence do y[J] := 0; 

 

{ --- calculate log factorials -------------------------------------- } 

 

                               Lfact[0] := 0; 

  for J:=1 to max_factorial do Lfact[J] := Lfact[J-1] + Ln( J ); 

 

{ ---  read data   -------------------------------------------------- } 

 

  write('Enter option (1.p-value 2:power 3.confidence set ==>> '); 

  readln( n_option ); 

 

       if n_option=2 then option := get_power 

  else if n_option=3 then option := get_CI 

  else                    option := get_p_value; 

 

  write('Enter data file name ==>> '); readln( filename ); 

 

  if FileExists( filename ) then 

    begin 

      assign( fi, filename);          reset(fi); 

      assign( fo, filename+'.csv' );  rewrite(fo); 

    end 

  else 

    begin 

      writeln; 

      writeln('Error : file NOT found. Please try it again!'); 

      goto last_line; 

    end; 

 

  J  := 0; 

  Ya := 0; 

 

  repeat 

    J := J + 1; 

    readln ( fi, y[J] ); 

    Ya := Ya + y[J]; 

  until eof(fi) or (J >= max_sequence); 

 

  a := J;  { --- a is the length of the observed sequence y[] --- } 

 

  if not eof(fi) then begin 

    writeln('Warning: the length of the sequence is greater than max.' ); 

    goto last_line; 

  end; 

 

  if (Ya > max_factorial) then begin 

    writeln('Warning: the sum of yk is greater than max(1754).' ); 

    goto last_line; 

  end; 

 

  close(fi); 

 



 

 

 

 

22                                                C. Hirotsu, H. Tsuruta 

  if (option=get_power) then begin 

    write('Enter change point ==>> ');     readln( change_point ); 

    write('Enter critical point ==>> ');   readln( critical_point ); 

  end; 

 

  writeln; 

  writeln( 'Now calulating ........ ' ); 

  writeln; 

 

{ ---  top level   -------------------------------------------------- } 

 

  case option of 

    get_p_value : 

      begin 

        p := 1 - get_F( a, 0, 0, 0, 0, y ); 

 

        writeln(     'p-value: ', p:10:6 ); 

        writeln( fo, 'p-value:,', p:10:6 ); 

      end; 

  

    get_power : 

      begin 

        writeln( '   delta        p' ); 

        writeln( '  ---------------------' ); 

        writeln( fo, 'delta,p' ); 

 

        for J:=-200 to 200 do begin 

          delta := J / 100; 

          p := 1 - get_F( a, 0, change_point, critical_point, delta, y ); 

 

          writeln(     delta:8:2, '  ', p:10:6 ); 

          writeln( fo, delta:8:2, ',',  p:10:6 ); 

        end; 

 

        writeln( '  ---------------------' ); 

      end; 

 

    get_CI : 

      begin 

        for K0:=1 to (a-1) do begin 

          p := 1 - get_F( a, K0, 0, 0, 0, y ); 

 

          if (K0=1) then begin 

            writeln( '      K0        p' ); 

            writeln( '    -----------------------' ); 

            writeln( fo, 'K0,p' ); 

          end; 

 

          write( K0:8, '  ', p:10:6 ); 

               if (p>=0.1)  then writeln( '   **' ) 

          else if (p>=0.05) then writeln( '   *' ) 

          else                   writeln; 

          write( fo, K0:8, ',',  p:10:6 ); 

               if (p>=0.1)  then writeln( fo, ',**' ) 

          else if (p>=0.05) then writeln( fo, ',*' ) 

          else                   writeln( fo ); 

        end; 

 

        writeln( '    -----------------------' ); 

      end; 

  end; 

 

{ ---  ending  ------------------------------------------------------ } 

 

  close(fo); 

 

last_line:  

  writeln; 

  write('Hit ENTER key !'); 

  readln; 

end. 

Worked Example 

   We give here only a short worked example since the real example of Table 1 

is too lengthy.  

For Examples 1 and 2 one should be recommended to visit our website. 

 

【Data File】 

1 

1 

1 

3 

3 

3 

 

【Output in the calculation of 𝑝-value】 

Enter option (1.p-value 2:power 3.confidence set ==>> 1 

Enter data file name ==>> PO_A 

Now calulating ........ 

     k     y[k]    Yk[k]     E[t]        V[t]        t[k] 

------------------------------------------------------------ 

     1       1       1      1.0000      1.6667      0.7746 

     2       1       2      1.0000      0.6667      1.2247 

     3       1       3      1.0000      0.3333      1.7321 

     4       3       6      0.5000      0.1667      1.2247 

     5       3       9      0.2000      0.0667      0.7746 

------------------------------------------------------------ 

max_t0 =    1.7321 at K =   3 

p-value:   0.147437 

Hit ENTER key ! 

 

【Output in the calculation of confidence set of change-point】 

Enter option (1.p-value 2:power 3.confidence set ==>> 3 

Enter data file name ==>> PO_A 

Now calulating ........ 

     k     y[k]    Yk[k]     E[t]        V[t]        t[k] 

------------------------------------------------------------ 

     1       1       1      1.0000      1.6667      0.7746 

     2       1       2      1.0000      0.6667      1.2247 

     3       1       3      1.0000      0.3333      1.7321 

     4       3       6      0.5000      0.1667      1.2247 

     5       3       9      0.2000      0.0667      0.7746 

------------------------------------------------------------ 

max_t0 =    1.7321 at K =   3 

      K0        p 

    ----------------------- 

       1    0.226435   ** 

       2    0.335275   ** 

       3    0.565521   ** 

       4    0.306808   ** 

       5    0.177867   ** 

    ----------------------- 

 

Hit ENTER key ! 


